Problem 25 in Section 16.3

Recall that in this problem we are studying the pyramid bounded by the planes $z = -6$, $y = 0$, $y - x = 4$ and $2x + y + z = 4$. In class we calculated all the vertices of this pyramid. Now I will calculate them in *Mathematica*

```
In[1]:= eqs = {z == -6, y == 0, y - x == 4, 2 x + y + z == 4}
Out[1]= {z == -6, y == 0, -x + y == 4, 2 x + y + z == 4}
```

Each vertex is at the intersection of three planes. So, I solve three of the four equations and get four vertices.

```
In[2]:= pP1 = {x, y, z} /. Solve[{z == -6, y == 0, y - x == 4}, {x, y, z}][[1]]
Out[2]= {-4, 0, -6}

In[3]:= pP2 = {x, y, z} /. Solve[{z == -6, y == 0, 2 x + y + z == 4}, {x, y, z}][[1]]
Out[3]= {5, 0, -6}

In[4]:= pP3 = {x, y, z} /. Solve[{z == -6, y - x == 4, 2 x + y + z == 4}, {x, y, z}][[1]]
Out[4]= {2, 6, -6}

In[5]:= pP4 = {x, y, z} /. Solve[{y == 0, y - x == 4, 2 x + y + z == 4}, {x, y, z}][[1]]
Out[5]= {-4, 0, 12}
```

Plot these points:

In[6]:=
```
Graphics3D[{
  {PointSize[0.02], Point[pP1], Point[pP2], Point[pP3], Point[pP4]},
  {Opacity[0.3], Polygon[{pP1, pP2, pP3}],
   Polygon[{pP1, pP2, pP4}], Polygon[{pP1, pP4, pP3}], Polygon[{pP4, pP2, pP3}]},
  {Thickness[0.01], Cyan, Line[{pP1, pP2}], Green, Line[{pP1, pP3}], Blue, Line[{pP2, pP3}],
   Red, Line[{pP1, pP4}], Magenta, Line[{pP2, pP4}], Yellow, Line[{pP3, pP4}]},
  {Text[P1, pP1, {1, 1}], Text[P2, pP2, {-1, 1}],
   Text[P3, pP3, {-1, -1}], Text[P4, pP4, {-1, -1}]}
 },
 PlotRange → {{-5, 6}, {-1, 8}, {-7, 13}},
 Axes → True, BoxRatios → {11, 9, 20}, AxesLabel → {x, y, z}]
```

Out[6]=



There are 6 lines of interest

Cyan

In[7]:=
```
liC = {x, y, z} /. Solve[{z == -6, y == 0}, {x, y, z}][[1]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

Out[7]=
```
{x, 0, -6}
```

Green

In[8]:=
```
liG = {x, y, z} /. Solve[{z == -6, y - x == 4}, {x, y, z}][[1]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

Out[8]=
```
{x, 4 + x, -6}
```

Blue

In[9]:=
```
liB = {x, y, z} /. Solve[{z == -6, 2 x + y + z == 4}, {x, y, z}][[1]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

Out[9]=
```
{x, 10 - 2 x, -6}
```

Red

In[10]:=
```
liR = {x, y, z} /. Solve[{y == 0, y - x == 4}, {x, y, z}][[1]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

Out[10]=
```
{-4, 0, z}
```

Magenta

In[11]:=
```
liM = {x, y, z} /. Solve[{y == 0, 2 x + y + z == 4}, {x, y, z}][[1]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

Out[11]=
```
{x, 0, 4 - 2 x}
```

Yellow

In[12]:=
```
liY = {x, y, z} /. Solve[{y - x == 4, 2 x + y + z == 4}, {x, y, z}][[1]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

Out[12]=
```
{x, 4 + x, -3 x}
```

■ How to integrate? (easier way)

How to integrate? Fix y to be y0 and find the corresponding points on Green, Blue and Yellow lines

In[13]:=
```
liG
```

Out[13]=
```
{x, 4 + x, -6}
```

In[14]:=
```
liB
```

Out[14]=
```
{x, 10 - 2 x, -6}
```

In[15]:=
```
liY
```

Out[15]=
```
{x, 4 + x, -3 x}
```

In[16]:=
```
liG /. Solve[liG[[2]] == y0, x][[1]]
```

Out[16]=
```
{-4 + y0, y0, -6}
```

In[17]:=
```
liB /. Solve[liB[[2]] == y0, x][[1]]
```

Out[17]=
$$\left\{\frac{10 - y0}{2}, y0, -6\right\}$$

In[18]:=
```
liY /. Solve[liY[[2]] == y0, x][[1]]
```

Out[18]=
```
{-4 + y0, y0, -3 (-4 + y0)}
```

In[19]:=
```
y0 = 2.5; pPG = {-4 + y0, y0, -6}; pPB = {(10 - y0)/2, y0, -6}; pPY = {-4 + y0, y0, -3 (-4 + y0)};
```

```
Graphics3D[{
  {PointSize[0.02], Point[pP1], Point[pP2], Point[pP3], Point[pP4]},
  {Opacity[0.3], Polygon[{pP1, pP2, pP3}],
   Polygon[{pP1, pP2, pP4}], Polygon[{pP1, pP4, pP3}], Polygon[{pP4, pP2, pP3}]},
  {Thickness[0.01], Cyan, Line[{pP1, pP2}], Green, Line[{pP1, pP3}], Blue, Line[{pP2, pP3}],
   Red, Line[{pP1, pP4}], Magenta, Line[{pP2, pP4}], Yellow, Line[{pP3, pP4}]},
  {Text[P1, pP1, {1, 1}], Text[P2, pP2, {-1, 1}],
   Text[P3, pP3, {-1, -1}], Text[P4, pP4, {-1, -1}]},
  {Polygon[{pPG, pPB, pPY}]}

 },
 PlotRange → {{-5, 6}, {-1, 8}, {-7, 13}},
 Axes → True, BoxRatios → {11, 9, 20}, AxesLabel → {x, y, z}]
```

Out[20]=



In[21]:=
```
Clear[y0]; Manipulate[pPG = {-4 + y0, y0, -6};
 pPB = {(10 - y0)/2, y0, -6}; pPY = {-4 + y0, y0, -3 (-4 + y0)};

 Graphics3D[{
   {PointSize[0.02], Point[pP1], Point[pP2], Point[pP3], Point[pP4]},
   {Opacity[0.3], Polygon[{pP1, pP2, pP3}],
    Polygon[{pP1, pP2, pP4}], Polygon[{pP1, pP4, pP3}], Polygon[{pP4, pP2, pP3}]},
   {Thickness[0.01], Cyan, Line[{pP1, pP2}], Green, Line[{pP1, pP3}],
    Blue, Line[{pP2, pP3}], Red, Line[{pP1, pP4}], Magenta,
    Line[{pP2, pP4}], Yellow, Line[{pP3, pP4}]}, {Text[P1, pP1, {1, 1}],
    Text[P2, pP2, {-1, 1}], Text[P3, pP3, {-1, -1}], Text[P4, pP4, {-1, -1}]},
   {Polygon[{pPG, pPB, pPY}]}

  },
  PlotRange → {{-5, 6}, {-1, 8}, {-7, 13}},
  Axes → True, BoxRatios → {11, 9, 20}, AxesLabel → {x, y, z}],
 {{y0, 2}, 0, 6}]
```

Out[21]=



In[22]:=
```
Integrate[Integrate[Integrate[1, {z, -6, 4 - 2 x - y}], {x, y - 4, (10 - y)/2}], {y, 0, 6}]
```

Out[22]= 162

How to integrate? (harder way)

How to integrate? Fix z to be z0 and find the corresponding points on Red, Magenta and Yellow lines

In[23]:=
```
liR
```

Out[23]= {-4, 0, z}

In[24]:=
```
liM
```

Out[24]= {x, 0, 4 - 2 x}

In[25]:=
```
liY
```

Out[25]= {x, 4 + x, -3 x}

In[26]:=
```
liR /. Solve[liR[[3]] == z0, z][[1]]
```

Out[26]= {-4, 0, z0}

In[27]:=
```
liM /. Solve[liM[[3]] == z0, x][[1]]
```

Out[27]= $\left\{\frac{4 - z0}{2}, 0, z0\right\}$

In[28]:=
```
liY /. Solve[liY[[3]] == z0, x][[1]]
```

Out[28]= $\left\{-\frac{z0}{3}, 4 - \frac{z0}{3}, z0\right\}$

In[29]:=
```
Clear[z0]; Manipulate[pPR = {-4, 0, z0}; pPM = {(4 - z0)/2, 0, z0}; pPY1 = {-z0/3, 4 - z0/3, z0};

 Graphics3D[{
   {PointSize[0.02], Point[pP1], Point[pP2], Point[pP3], Point[pP4]},
   {Opacity[0.3], Polygon[{pP1, pP2, pP3}],
    Polygon[{pP1, pP2, pP4}], Polygon[{pP1, pP4, pP3}], Polygon[{pP4, pP2, pP3}]},
   {Thickness[0.01], Cyan, Line[{pP1, pP2}], Green, Line[{pP1, pP3}],
    Blue, Line[{pP2, pP3}], Red, Line[{pP1, pP4}], Magenta,
    Line[{pP2, pP4}], Yellow, Line[{pP3, pP4}]}, {Text[P1, pP1, {1, 1}],
    Text[P2, pP2, {-1, 1}], Text[P3, pP3, {-1, -1}], Text[P4, pP4, {-1, -1}]},
   {Thickness[0.01], Cyan, Line[{pPR, pPM}], Green,
    Line[{pPR, pPY1}], Blue, Line[{pPM, pPY1}]},
   {PointSize[0.02], Point[pPR], Point[pPM], Point[pPY1]},
   {Polygon[{pPR, pPM, pPY1}]}

  },
  PlotRange → {{-5, 6}, {-1, 8}, {-7, 13}},
  Axes → True, BoxRatios → {11, 9, 20}, AxesLabel → {x, y, z}],
 {{z0, 2}, -6, 12}]
```

Out[29]=



Recall

In[30]:=
```
pPY1 = {-z0/3, 4 - z0/3, z0}
```

Out[30]=
$$\left\{-\frac{z0}{3},\ 4-\frac{z0}{3},\ z0\right\}$$

So, y is between 0 and $4 - \frac{z}{3}$.

Recall, the green and blue line, but now change them to be at the level z0

Green

In[31]:=
```
liGz = {x, y, z} /. Solve[{z == z0, y - x == 4}, {x, y, z}][[1]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

Out[31]=
```
{x, 4 + x, z0}
```

Blue

In[32]:=
```
liBz = {x, y, z} /. Solve[{z == z0, 2 x + y + z == 4}, {x, y, z}][[1]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

Out[32]=
```
{x, 4 - 2 x - z0, z0}
```

In[33]:=
```
Integrate[Integrate[Integrate[1, {x, y - 4, (4 - y - z)/2}], {y, 0, 4 - z/3}], {z, -6, 12}]
```

Out[33]=
```
162
```